
Notice d’emploi

Pour une somme empruntée et remboursée en une fois : 

vc =.tp vcap va ||   ta =.va taux vc
va =.tp vact vc  ||   ta =.np tann tp    
tp =.np tper ta      || 

vc : valeur capitalisée (remboursée) va : valeur actualisée (empruntée)
 ta : taux annuel (ou nominal)     tp : taux périodique 
np : nombre de périodes (ou échéances) par an. (mensuel, trimestriel…)
 
Pour une rente ou remboursements échelonnés :

  vcr =.tp vcap_rente ech  ||  var=.tp vact_rente ech

vcr : valeur capitalisée d’une rente            var : valeur actualisée d’une rente
ech : vecteur des échéances constantes ou non 
       
ech_c =.(tp rente_cste ne)va ||  tp =.va tp_rente ech

ech_c : échéance constante                         ne : nombre d’échéances
ech : vecteur des échéances constantes ou non 

Tableau des échéances avec détails capital et intérêt :

  tab_echeances =.ne (np echeancier ta) va 
      
ta : taux annuel (ou nominal) ne : nombre d’échéances
np : nombre de périodes (ou nombre d’échéances) par an
période : intervalle de temps constant séparant 2 échéances
     (   1 : période = année    ;    2 : période = 6 mois )
     (   3 : période = 4 mois   ;    4 : période = 3 mois )
     (   6 : période = 2 mois   ; 12 : période = 1 mois )
tab_echeances : tableau de 5 colonnes constituant un échéancier
       col 1 (No) : numéro de l’échéance 
       col 2 (Ve) : valeur de l’échéance constante
       col 3 (I   ) : part d’intérêt composant une échéance
       col 4 (C  ) : part de capital composant une échéance
       col 5 (Vr) : valeur de remboursement (ou valeur résiduelle) 

                                   nae    =.   (  va     nb_ech     ech_c)     tp 
 

nae : nombre approché d’échéances constantes (on a ne =. partie entière de nae) 
Il faut ajouter une échéance de valeur inférieure.



Détails de programmation

NB. valeur capitalisée :
   vcap =: ]*[:>:[
NB. calcul du taux :
   taux =: _1+%~
NB. taux périodique
   tper =: _1+(1+])^[:%[

NB. valeur actualisée
   vact =: ]%[:>:[
NB. taux annuel
   tann =: _1+[^~1+]

NB. val. cap.  Rente
 vcap_rente =: [:+/]*(1+[)^[:|.[:i.[:#]
NB. val. act. rente
 vact_rente =: [:+/]%(1+[)^1+[:i.[:#]  
NB. échéance cste
 rente_cste =: 2 : 'y*m%1-(1+m)^-n'

  NB. Calcul du taux périodique d’une rente  d’échéances constantes ou non :
 tp_rente    =: 3 : 0  
[:  
:
i=.0[N=.#y[t=.0.05
while.  16>i=.>:i do.
t=.t+((+/y%(1+t)^1+i.N)-x)%+/y*(1+i.N)%(1+t)^2+i.N
end.
t
) 

Remarque :  On utilise ici la formule de NEWTON permettant le calcul
d’une racine d’une équation. En partant de la valeur initiale 0.05                      
l’expérience prouve que 16 itérations suffisent pour atteindre la précision limite 
(la valeur 16 peut être modifiée dans le « while. »).

NB. calcul des éléments du tableau d’un échéancier :
echeancier =: 2 : 0 

[:
:
r =.((t=.m tper n) rente_cste x)y
c =.r-i=.t*y,}:V=.y*1-(_1+(1+t)^j=.1+i.x)%_1+(1+t)^x
s =. 'No';' Ec';' I';' C';' Vr'
z =. (U f c);(U=.3j2+>.10^.y)(f=.":,.) V
s,:(3 f j);(U f x$r);(U f i);z
)
Remarque : f est un pro-verbe local (sous forme tacite) inconnu hors exécution 
de la pro-conjonction   « echeancier ».

NB. Nombre approché d’échéances :
nb_ech =: 2 : '-(1+y)^.1-m*y%n' 



Exemples d’utilisation 
1) Taux semestriel (2 échéances par an) quand le taux annuel est 8% :
2 tper 0.08

0.0392305
2) Taux annuel correspondant à un taux mensuel de 1% :
12 tann 0.01

0.126825  
3) Taux mensuel correspondant à  un taux trimestriel de 2% :
12 tper 4 tann 0.02

0.00662271
4) Taux annuel de 10000 € remboursés en 5 échéances semestrielles 

(2 échéances par an)   non constantes  (  2000  1000  3000  4000  2000  ) :
2 tann 10000 tp_rente 2000 1000 3000 4000 2000

0.120641         NB. C’est le taux annuel  (ou nominal)  pratiqué
5) Échéancier pour 1000000 € remboursés en 12 échéances trimestrielles 

constantes (4 versements par an), au taux annuel  (ou nominal)  de 5,5 % :
12  (4 echeancier 0.055) 1000000

┌───┬─────────┬─────────┬─────────┬─────────┐
│No │ Ec      │ I       │ C       │ Vr      │
├───┼─────────┼─────────┼─────────┼─────────┤
│  1│ 90811.42│ 13475.17│ 77336.25│922663.75│
│  2│ 90811.42│ 12433.05│ 78378.37│844285.38│
│  3│ 90811.42│ 11376.89│ 79434.53│764850.85│
│  4│ 90811.42│ 10306.50│ 80504.93│684345.93│
│  5│ 90811.42│  9221.68│ 81589.74│602756.18│
│  6│ 90811.42│  8122.24│ 82689.18│520067.00│
│  7│ 90811.42│  7007.99│ 83803.43│436263.57│
│  8│ 90811.42│  5878.73│ 84932.70│351330.88│
│  9│ 90811.42│  4734.24│ 86077.18│265253.70│
│ 10│ 90811.42│  3574.34│ 87237.08│178016.61│
│ 11│ 90811.42│  2398.80│ 88412.62│ 89603.99│
│ 12│ 90811.42│  1207.43│ 89603.99│     0.00│
└───┴─────────┴─────────┴─────────┴─────────┘

NB. Vérification :       
4  tann 1000000 tp_rente  12 $ 90811.42      

0.055             NB. C’est tout bon !

6) Pour payer des meubles de 8000 € en 12 mensualités, un vendeur 
propose d’ajouter 11,5% et de diviser par 12. Quel est le « vrai » taux annuel ?
    12 tann va tp_rente 12$12%~(va=.8000)*(1+0.115)
0.226475      NB.  Plus de 22,6% ! Si ce n’est pas de l’usure, ça y ressemble !

7) Prêt de 10000 €, 18 mensualités constantes, taux annuel « avoué » 8% ,
« frais de dossier » 400 € payables avec la 1re échéance. Quel est le vrai taux ?

r=.((12 tper 0.08) rente_cste 18) 10000
    12 tann 10000 tp_rente (r+400),17$r
0.137956      NB. Taux annuel de presque 13,8%  et  non 8%    ! ! ! ! ! !

